
5 POKE36879,26::PRINT" "

10 REM SET VARIABILI
11 SX=11:SY=12:DD=62:SC=0:CT=0:TX=SX:TY=SY

19 REM CALCOLO E POSIZIONAMENTO TOPOLINI
20 FORA=1TO5
22 :CX=RND(0)*23:CY=RND(0)*22
24 :POKE7680+CX+CY*22,42:POKE38400+CX+CY*22,0
26 NEXT

29 REM POSIZIONAMENTO SERPENTE
30 POKE7680+SX+SY*22,62:POKE38400+SX+SY*22,0

49 REM GESTIONE EVENTI TASTIERA
50 GETA$
51 IFA$="S"THENDD=62
52 IFA$="A"THENDD=60
53 IFA$="W"THENDD=65
54 IFA$="Z"THENDD=83
55 POKEX,DD:POKEX+30720,0
60 IFDD=62THENSX=SX+1
62 IFDD=60THENSX=SX-1
64 IFDD=65THENSY=SY-1
66 IFDD=83THENSY=SY+1

70 REM CONTROLLO COLLISIONE BORDI
71 IFSX<0ORSX>=22THEN2000
72 IFSY<0ORSY>22THEN2000

79 REM COLLISIONE CON ELEMENTI DI GIOCO
80 X=7680+SX+SY*22
82 V=PEEK(X)
84 IFV=42THENGOSUB1000
86 IFV<>32ANDV<>42THEN2000

87 REM GESTIONE CODA SERPENTE
88 IFCT>0THENCT=CT-1:GOTO50
90 TT=7680+TX+TY*22:VV=PEEK(TT)
95 POKETT,32:POKETT+30720,0
100 IFVV=62THENTX=TX+1
101 IFVV=60THENTX=TX-1
102 IFVV=65THENTY=TY-1
103 IFVV=83THENTY=TY+1
105 GOTO50

999 REM MANGIO TOPOLINO
1000 SC=SC+1:CT=5
1002 CX=RND(0)*22:CY=RND(0)*21
1003 G=7680+CX+CY*22:V=PEEK(G)
1004 IFV<>32THEN1002
1006 POKEG,42:POKEG+30720,0
1008 RETURN

1999 REM COLLISIONE!!
2000 PRINT"HAI PERSO!!":PRINT"PUNTI: "SC
2001 END


